
Sommario

Nello studio delle orbite ellittiche i tradizionali elementi (M,a,e,i,ω,Ω) presentano dei casi di
indeterminazione, e quindi di impossibilità risolutiva, quando l’orbita ha inclinazione nulla
(i=0) e/o circolare (e=0). Per superare qualsiasi ambiguità gli astronomi hanno introdotto
in meccanica celeste le Variabili Ellittiche, che risultano di grande efficacia e precisione nel
calcolo della posizione e della velocità di un corpo celeste. I relativi algoritmi, corredati del
listato di un programma BASIC, sono descritti in questo lavoro.

1 Introduzione

L’uso delle Variabili Ellittiche consente di definire un’orbita, cioè di calcolarne i vettori posi-
zione e velocità, anche nelle situazioni in cui i processi numerici di alcune applicazioni possono
essere instabili. Ciò si verifica quando l’inclinazione i → 0, o l’eccentricità e → 0, o nelle loro
combinazioni.

Linea
dei nodi

Perielio

2 Le Variabili Ellittiche

Le relazioni che legano questi nuovi parametri a quelli comunemente usati nella generalità dei casi
(M,a,e,i,ω,Ω) sono le seguenti: '

&

$

%

k= e · cos$
h= e · sin$

q= sin i

2
· cosΩ

p= sin i

2
· sinΩ

L= Longitudine media

a= Semiasse maggiore

essendo $ la longitudine del perielio, (e) l’eccentricità, Ω la longitudine del nodo ed (i) l’inclinazione
dell’orbita.
L’adozione di un algoritmo generale che non degeneri qualunque sia la forma e la posizione del-
l’orbita è di primaria importanza nello studio del calcolo orbitale.
Dalla meccanica celeste sappiamo, e lo si vede anche in figura, che la longitudine del perielio vale
$ = Ω+ ω , mentre la longitudine media si ottiene aggiungendo alla longitudine del perielio l’a-

nomalia media M, ovvero L = M +$ .
Nella tabella seguente riportiamo le varie combinazioni tra inclinazione ed eccentricità nulle poste
in relazione ai tre parametri orbitali (M,ω,Ω), che in ciascun caso possono essere indeterminati

e/o noti.

M ω Ω Caso
i = 0 e 6= 0 nota indeterminata indeterminata A
i 6= 0 e = 0 indeterminata indeterminata nota B
i = 0 e = 0 indeterminata indeterminata indeterminata C
i 6= 0 e 6= 0 nota nota nota D

Il caso (A) rappresenta un moto ellittico che avviene sull’eclittica. Dalle osservazioni (o dalle
tabelle) deve essere nota la longitudine del peielio $ e di conseguenza la longitudine media L,
essendo conosciuta M. Poichè sin i

2
= 0, le variabili (q, p) sono nulle per qualsiasi valore di Ω e

pertanto il moto risulta definito.

Nel caso (B) siamo in presenza di un moto circolare uniforme e inclinato rispetto all’eclittica;
la longitudine media L, che coincide con quella vera `, deve essere nota dalle tabelle. Le variabili
(k, h) sono uguali a zero, (q, p) completamente definite e il moto risulta calcolabile.

(C) è chiaramente il moto circolare che si sviluppa sull’eclittica; le quattro variabili (k, h, q, p)
sono nulle, è nota L = l ed il moto risulta definito.

Infine, nel caso generale (D), il moto è ellittico inclinato, tutte le tre grandezze (M,ω,Ω) so-
no determinate e le variabili ellittiche si calcolano con le formule riportate nei riquadri più in alto.

3 Il Vettore Posizione

In figura sono disegnati sia gli assi eclittici eliocentrici (x,y,z) con l’asse x diretto verso il punto
d’Ariete γ che quelli orbitali (ξ, η, ζ), la cui direzione principale ξ è rivolta verso il perielio. Un’altra
terna d’assi, anch’essa orbitale (ξ ′, η′, ζ ′), viene definita tramite una rotazione di (-$) nel piano
orbitale con l’asse ζ ′ coincidente con ζ.
Con opportune manipolazioni algebriche, che esulano da questa trattazione, il vettore posizione

del punto P riferito agli assi principali eclittici si calcola con la formula seguente, ripresa da p. 311
di [1]:











x = K1 · r cos `+K0 · r sin `
y = K2 · r sin `+K0 · r cos `
z = −K3 · r cos `+K4 · r sin `

essendo K0 = 2pq, K1 = (1 − 2p2), K2 = (1 − 2q2), K3 = 2p
√
1 − p2 − q2, K4 = 2q

√
1 − p2 − q2.

La grandezza ` rappresenta la longitudine vera del corpo e si calcola sommando la longitudine del
perielio all’anomalia vera θ, cioè ` = θ +$ .
L’anomalia θ è l’incognita del problema e si calcola attraverso una nuova forma dell’equazione di

Keplero, che è la seguente:
L = F + h · cosF − k · sinF

dove F, detta longitudine eccentrica, si determina risolvendo l’equazione per successive approssi-
mazioni.
Nota F, si determina l’anomalia eccentrica E = F − $ e quindi il raggio vettore r = a(1 − e cosE) .

L’anomalia vera (θ) si ricava con la nota formula:

tan
θ

2
=

√

1 + e

1 − e
· tan E

2

per cui, essendo a questo punto conosciuta ` = θ+$, il vettore posizione (x, y, z) è completamente
definito.

4 Il Vettore Velocità

Si ottiene derivando rispetto al tempo il vettore posizione, cioè calcolando ẋ = dx/dt = Vx,
ẏ = dy/dt = Vy, ż = dz/dt = Vz.
Le formule sono le seguenti:











Vx = K1 · [r cos `]′ +K0 · [r sin `]′ = K1 · V ′
c +K0 · V ′

s

Vy = K2 · [r sin `]′ +K0 · [r cos `]′ = K2 · V ′
s +K0 · V ′

c

Vz = −K3 · [r cos `]′ +K4 · [r sin `]′ = −K3 · V ′
c +K4 · V ′

s

Calcoliamo le due derivate V ′
c e V ′

s , che, in base ai noti teoremi meccanici sul moto centrale,
risultano essere:

V ′
c =

√

µ

p̃
· [e sin θ cos ` − (1 + e cos θ) sin `]

V ′
s =

√

µ

p̃
· [e sin θ sin `+ (1 + e cos θ) cos `]

essendo p̃ = a(1 − e2) il parametro orbitale.

5 Il listato BASIC

’^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^

’ Programma: VAR_ELLI.BAS

’ Soggetto: Calcolo dei vettori Posizione e Velocita’ di un corpo P

’ usando le Variabili Ellittiche.

’ L’algoritmo e’ sempre valido anche quando i=0 e quindi il

’ Nodo (Om) e l’Argom. del perielio (om) sono INDETERMINATI

’ Data : 3-11-2003

’^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^

COLOR 14, 1: CLS

DEFDBL A-Z

pi = 4 * ATN(1#): rad = pi / 180

DEF fnc (x) = x + h * COS(x) - k * SIN(x) - L: ’Funzione

DEF fnc1 (x) = 1 - h * SIN(x) - k * COS(x): ’Derivata prima

DEF FNridu (x) = (x / (2 * pi) - (INT(x / (2 * pi)) - (x < 0))) * 2 * pi

DEF FNneg (x) = x - (x < 0) * 2 * pi

’

’-------- Dati del problema -----------

omseg = 268.75092#

a = .756656#

e = .3978305#

L = 35.63053

i = 6.87631

Nodo = 63.07572

’--------------------------------------

PRINT : PRINT SPC(20); "Dati del Problema"

PRINT USING "Long.Media= ###.##### gradi Long.Perielio= ###.##### gradi";L;omseg

PRINT USING "Semias.mag= ###.####### UA Eccentric.= #.#######"; a; e

PRINT USING " Inclinaz.= ##.##### gradi Long.Nodo= ###.##### gradi"; i; Nodo

’

’ Angoli in RAD

omseg = omseg * rad

L = L * rad

i = i * rad

Nodo = Nodo * rad

’------------------- Inizio calcoli -----------------

k = e * COS(omseg)

h = e * SIN(omseg)

q = SIN(i / 2) * COS(Nodo)

p = SIN(i / 2) * SIN(Nodo)

K1 = 1 - 2 * p ^ 2

K2 = 1 - 2 * q ^ 2

K3 = 2 * p * SQR(1 - p ^ 2 - q ^ 2)

K4 = 2 * q * SQR(1 - p ^ 2 - q ^ 2)

K0 = 2 * p * q

PRINT : PRINT SPC(10); "Equaz. di Keplero risolta per successive approssimaz."

’

’ Equazione di Keplero per variabili ellittiche: L=F+h*COS(F)-k*SIN(F)

’ ---------------------

x = L: ’Valore iniziale

DO

variaz = -fnc(x) / fnc1(x)

kount = kount + 1

x = x + variaz

PRINT kount, x, variaz

LOOP UNTIL ABS(variaz) < 1E-12

PRINT : PRINT SPC(10); "Grandezze intermedie"

PRINT USING " F= #.######### rad = ###.###### gradi"; x; x / rad

Eg = FNneg(FNridu(x - omseg)): ’ essendo F=E+omseg

r = a * (1 - e * COS(Eg))

PRINT USING " E= ####.###### gradi r= #.######## UA"; Eg / rad; r

’ Calcolo di (teta)

teta = 2 * ATN(TAN(Eg / 2) * SQR((1 + e) / (1 - e)))

elle = FNridu(teta + omseg)

PRINT USING " te= ###.###### gradi l= ###.###### gradi"; teta/rad; elle/rad

’

’ Vettore POSIZIONE

x = K1 * r * COS(elle) + K0 * r * SIN(elle)

y = K2 * r * SIN(elle) + K0 * r * COS(elle)

z = -K3 * r * COS(elle) + K4 * r * SIN(elle)

’------------ Coo. Equatoriali -----------’

’ eps = 23.4392911#: ’obliquita’ eclittica

’ cc = COS(eps * rad): ss = SIN(eps * rad)

’ yeq = cc * y - ss * z: zeq = ss * y + cc * z

’ y = yeq: z = zeq

’-----------------------------------------’

PRINT : PRINT SPC(13); "Risultati: Vettori POSIZIONE e VELOCITA’"

PRINT USING " x= ##.######## y= ##.######## z= ##.########

-> r= #.######## UA"; x; y; z; r

’ Vettore VELOCITA’

pp = a * (1 - e ^ 2): ’ parametro orbitale

VpriC = SQR(1 / pp) * (e*SIN(teta) * COS(elle) - (1+e*COS(teta)) * SIN(elle))

VpriS = SQR(1 / pp) * (e*SIN(teta) * SIN(elle) + (1+e*COS(teta)) * COS(elle))

’

Vx = K1 * VpriC + K0 * VpriS

Vy = K2 * VpriS + K0 * VpriC

Vz = -K3 * VpriC + K4 * VpriS

V = SQR(Vx ^ 2 + Vy ^ 2 + Vz ^ 2)

’---------------- Vel. Equatoriali -------------------’

’ Vyeq = cc * Vy - ss * Vz: Vzeq = ss * Vy + cc * Vz

’ Vy = Vyeq: Vz = Vzeq

’-----------------------------------------------------’

PRINT USING " Vx= ##.######## Vy= ##.######## Vz= ##.########

-> V= #.######## UA/UT0"; Vx; Vy; Vz; V

END

’*********** Fine Programma: VAR_ELLI.BAS ******** (Nov.03) ***********

6 Esempio

Gli elementi orbitali si riferiscono all’asteroide 1994 WR12 (Near Earth Object). La sua posizione
e velocità è calcolata all’epoca 25/11/1994, 0.00 TU; quest’ultima è misurata in unità canonica
UA/UT0, che si trasforma in km/sec moltiplicandola per 29.78469.

Dati del Problema

Long.Media= 35.63053 gradi Long.Perielio= 268.75092 gradi

Semias.mag= 0.7566560 UA Eccentric.= 0.3978305

Inclinaz.= 6.87631 gradi Long.Nodo= 63.07572 gradi

Equaz. di Keplero risolta per successive approssimaz.

1 .8787607807968485 .2568906951843532

2 .8713126307131766 -7.448150083671901D-03

3 .8713073815918247 -5.249121351832862D-06

4 .8713073815892014 -2.623391652023067D-12

5 .8713073815892014 -4.56468862800706D-17

Grandezze intermedie

F= 0.871307382 rad = 49.922236 gradi

E= 141.171316 gradi r= 0.99115850 UA

te= 153.950828 gradi l= 62.701748 gradi

Risultati: Vettori POSIZIONE e VELOCITA’

x= 0.45452605 y= 0.88079547 z= -0.00077455 -> r= 0.99115850 UA

Vx= -0.60995560 Vy= 0.56118671 Vz= 0.09622809 -> V= 0.83440769 UA/UT0

